Ethereum入門　〜Mistを使って独自トークンの発行〜

#はじめに
前回browser-solidityを使って独自トークンを発行しました。
[Ethereum入門　〜独自トークンの発行〜](https://qiita.com/kolife/items/43926fcfde2d6a392aaf)
今回はEthereumの公式ウォレットのMistを使って送金、独自トークン発行までしたいと思います。

#Mistインストール・起動
Mistのダンロードはこちらから
https://github.com/ethereum/mist/releases

OSに対応したMistをインストール
<img width="772" alt="スクリーンショット 2018-01-24 11.09.57.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/d92fcd93-0e1c-e1e5-795d-878f29c235a0.png">

起動まではこちらに丁寧な解説あります。
[Ethereum ＜超初心者の方向け＞Mistを用いたコントラクト開発入門](http://pinkworks.hatenablog.com/entry/2017/12/12/004815)

<img width="394" alt="スクリーンショット 2018-01-24 10.04.38.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/c48885af-97e3-c7f1-4436-02287e8d2cab.png">

この画面になったらLAUNCH APPLICATION を押して起動

#送金

###アカウント作成
まずはアカウントを2つ作成します。
1つ目に作ったアカウントがEtherbaseとなりマイニングすると報酬がもらえるアカウントになります。

<img width="450" alt="スクリーンショット 2018-01-24 10.09.43.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/be54a1ed-926a-7d49-4088-53b8c5fd9583.png">

<img width="1045" alt="スクリーンショット 2018-01-24 10.14.55.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/d2d3a979-bef0-ac00-80dc-a8f352c73ffd.png">

###マイニング開始
Develop→マイニングを始める
<img width="276" alt="スクリーンショット 2018-01-24 11.00.01.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/6ce1d0b9-6ca3-3f4e-fead-ebad0865743a.png">

1つ目に作ったアカウントにEthがどんどん溜まっていきます。

###送金
アカウント1からアカウント2へ100Eth送金します。
<img width="1048" alt="スクリーンショット 2018-01-24 10.15.52.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/8e19d85b-944a-d133-a765-e29ea96b5c55.png">
Toと量を入力して「送る」ボタンを押します。

<img width="1049" alt="スクリーンショット 2018-01-24 10.16.28.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/d742e129-5431-5729-064d-771847c35b8f.png">

無事にアカウント2に100Eth送金されました。

トランザクションの確認をすると
ガス、ブロックを確認することが出来ます。
<img width="561" alt="スクリーンショット 2018-01-24 10.25.10.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/057a400c-84cd-04a8-85fe-00ea73a115a6.png">


###マイニング終了
マイニング終了するためにはコンソールからコマンドを叩きます。
Develop　→　開発者用ツールをトグルする　→　MistUI

<img width="613" alt="スクリーンショット 2018-01-24 11.00.18.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/a2b37c91-7a37-bef9-be3b-bc4a036ba2b7.png">

以下のコマンドを入力してマイニングを停止します。
```
web3.miner.stop()
```

<img width="1047" alt="スクリーンショット 2018-01-24 10.22.55.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/777f2fc2-b4ec-5c2b-b5a8-37c3e0c0f1d7.png">


#トークンの発行
###コントラクト作成
<img width="1051" alt="スクリーンショット 2018-01-24 10.34.53.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/40dc7994-5921-bcb9-3e44-f2dbcf282472.png">

コードはこちら
https://ethereum.org/token

```
pragma solidity ^0.4.16;

interface tokenRecipient { function receiveApproval(address _from, uint256 _value, address _token, bytes _extraData) public; }

contract TokenERC20 {
    // Public variables of the token
    string public name;
    string public symbol;
    uint8 public decimals = 18;
    // 18 decimals is the strongly suggested default, avoid changing it
    uint256 public totalSupply;

    // This creates an array with all balances
    mapping (address => uint256) public balanceOf;
    mapping (address => mapping (address => uint256)) public allowance;

    // This generates a public event on the blockchain that will notify clients
    event Transfer(address indexed from, address indexed to, uint256 value);

    // This notifies clients about the amount burnt
    event Burn(address indexed from, uint256 value);

    /**
     * Constrctor function
     *
     * Initializes contract with initial supply tokens to the creator of the contract
     */
    function TokenERC20(
        uint256 initialSupply,
        string tokenName,
        string tokenSymbol
    ) public {
        totalSupply = initialSupply * 10 ** uint256(decimals);  // Update total supply with the decimal amount
        balanceOf[msg.sender] = totalSupply;                // Give the creator all initial tokens
        name = tokenName;                                   // Set the name for display purposes
        symbol = tokenSymbol;                               // Set the symbol for display purposes
    }

    /**
     * Internal transfer, only can be called by this contract
     */
    function _transfer(address _from, address _to, uint _value) internal {
        // Prevent transfer to 0x0 address. Use burn() instead
        require(_to != 0x0);
        // Check if the sender has enough
        require(balanceOf[_from] >= _value);
        // Check for overflows
        require(balanceOf[_to] + _value > balanceOf[_to]);
        // Save this for an assertion in the future
        uint previousBalances = balanceOf[_from] + balanceOf[_to];
        // Subtract from the sender
        balanceOf[_from] -= _value;
        // Add the same to the recipient
        balanceOf[_to] += _value;
        Transfer(_from, _to, _value);
        // Asserts are used to use static analysis to find bugs in your code. They should never fail
        assert(balanceOf[_from] + balanceOf[_to] == previousBalances);
    }

    /**
     * Transfer tokens
     *
     * Send `_value` tokens to `_to` from your account
     *
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transfer(address _to, uint256 _value) public {
        _transfer(msg.sender, _to, _value);
    }

    /**
     * Transfer tokens from other address
     *
     * Send `_value` tokens to `_to` on behalf of `_from`
     *
     * @param _from The address of the sender
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transferFrom(address _from, address _to, uint256 _value) public returns (bool success) {
        require(_value <= allowance[_from][msg.sender]);     // Check allowance
        allowance[_from][msg.sender] -= _value;
        _transfer(_from, _to, _value);
        return true;
    }

    /**
     * Set allowance for other address
     *
     * Allows `_spender` to spend no more than `_value` tokens on your behalf
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     */
    function approve(address _spender, uint256 _value) public
        returns (bool success) {
        allowance[msg.sender][_spender] = _value;
        return true;
    }

    /**
     * Set allowance for other address and notify
     *
     * Allows `_spender` to spend no more than `_value` tokens on your behalf, and then ping the contract about it
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     * @param _extraData some extra information to send to the approved contract
     */
    function approveAndCall(address _spender, uint256 _value, bytes _extraData)
        public
        returns (bool success) {
        tokenRecipient spender = tokenRecipient(_spender);
        if (approve(_spender, _value)) {
            spender.receiveApproval(msg.sender, _value, this, _extraData);
            return true;
        }
    }

    /**
     * Destroy tokens
     *
     * Remove `_value` tokens from the system irreversibly
     *
     * @param _value the amount of money to burn
     */
    function burn(uint256 _value) public returns (bool success) {
        require(balanceOf[msg.sender] >= _value);   // Check if the sender has enough
        balanceOf[msg.sender] -= _value;            // Subtract from the sender
        totalSupply -= _value;                      // Updates totalSupply
        Burn(msg.sender, _value);
        return true;
    }

    /**
     * Destroy tokens from other account
     *
     * Remove `_value` tokens from the system irreversibly on behalf of `_from`.
     *
     * @param _from the address of the sender
     * @param _value the amount of money to burn
     */
    function burnFrom(address _from, uint256 _value) public returns (bool success) {
        require(balanceOf[_from] >= _value);                // Check if the targeted balance is enough
        require(_value <= allowance[_from][msg.sender]);    // Check allowance
        balanceOf[_from] -= _value;                         // Subtract from the targeted balance
        allowance[_from][msg.sender] -= _value;             // Subtract from the sender's allowance
        totalSupply -= _value;                              // Update totalSupply
        Burn(_from, _value);
        return true;
    }
}

```

###コントラクト実行
「Token　ERC20」のコントラクト選択し、トークンの名前、トークンシンボルを入力します。今回はOreOreCoinを10000　発行します。

<img width="1051" alt="スクリーンショット 2018-01-24 10.34.53.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/40dc7994-5921-bcb9-3e44-f2dbcf282472.png">

「配置」ボタンを押し、アカウントのパスワードを入力し、コントラクトを実行します。

<img width="561" alt="スクリーンショット 2018-01-24 10.35.37.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/fea26a9b-a90e-97a6-51fe-d1fb234a9f51.png">

トランザクションを確認しコントラクトが実行出来ていればOK。
(実行出来ていない場合は、マイニングが実行されているか確認してみてください。)

<img width="579" alt="スクリーンショット 2018-01-24 10.37.24.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/645db025-cbe6-8c72-ded1-3c2b890054f4.png">

無事OreOreCoinが10000　OCアカウント2に発行されました。
<img width="918" alt="スクリーンショット 2018-01-24 10.38.09.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/9e7abec8-0d03-7be3-5d9b-232f5f86d50e.png">

###トークンの送金
アカウント2からアカウント3へトークンを送金してみます。
まずコントラクト内のselect functionでtransferを選択。
toにアカウント3のアドレスを入力し、valuに送る数量を入力します。
単位がweiなので注意！
1000送金したい場合は1000e18と入力します。

<img width="900" alt="スクリーンショット 2018-01-24 11.40.58.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/98ba75a9-0117-0ff0-6f44-380fca5533e6.png">

実行を押し、アカウントのパスワードを入力してSEND TRANSACTIO！

<img width="565" alt="スクリーンショット 2018-01-24 10.40.08.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/4d360e1b-b981-4a8e-e0f8-c08ef2a1bef6.png">


アカウント2から３へ1000　OC送金出来ました。
<img width="778" alt="スクリーンショット 2018-01-24 10.40.53.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/f853b41c-36cf-9b81-137f-4f254383616e.png">

<img width="778" alt="スクリーンショット 2018-01-24 10.41.05.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/90a29115-e9a8-f6a6-429e-d605d00945a8.png">

###トークンのバーン
バーンも同様の手順で
コントラクト内のselect functionでburnを選択。valuを入力し実行。

<img width="561" alt="スクリーンショット 2018-01-24 10.42.09.png" src="https://qiita-image-store.s3.amazonaws.com/0/200669/9da2331f-93d7-db17-6194-a2f3a5912eef.png">

ちなみにバーンはコントラクトのアドレスに送金することで誰も取り出せなくなるようにしてます。

#おわりに
Mistはシンプルで直感的で分かりやすいですね。
gethのインストール等々が不要なのでエンジニア以外も触れやすそうです。

#参考にさせて頂きました。
[Create your own CRYPTO-CURRENCY with Ethereum](https://ethereum.org/token)
[Ethereum ＜超初心者の方向け＞Mistを用いたコントラクト開発入門](http://pinkworks.hatenablog.com/entry/2017/12/12/004815)
[EthereumでERC-20トークンを発行してみる](http://ykubot.com/2017/12/30/ethereum-erc-20-token/)